



ΠΑΝΕΠΙΣΤΗΜΙΟ ΙΩΑΝΝΙΝΩΝ  
ΑΝΟΙΚΤΑ ΑΚΑΔΗΜΑΪΚΑ ΜΑΘΗΜΑΤΑ



# Υπολογιστές I

Συναρτήσεις

Διδάσκοντες: **Αν. Καθ. Δ. Παπαγεωργίου,**  
**Αν. Καθ. Ε. Λοιδωρικής**



Ευρωπαϊκή Ένωση  
Ευρωπαϊκό Κοινωνικό Ταμείο  
ΕΠΙΧΕΙΡΗΣΙΑΚΟ ΠΡΟΓΡΑΜΜΑ  
ΕΚΠΑΙΔΕΥΣΗ ΚΑΙ ΔΙΑ ΒΙΟΥ ΜΑΘΗΣΗ  
ΥΠΟΥΡΓΕΙΟ ΠΑΙΔΕΙΑΣ & ΘΡΗΣΚΕΥΜΑΤΩΝ, ΠΟΛΙΤΙΣΜΟΥ & ΑΘΛΗΤΙΣΜΟΥ  
ΕΙΔΙΚΗ ΥΠΗΡΕΣΙΑ ΔΙΑΧΕΙΡΙΣΗΣ  
Με τη συγχρηματοδότηση της Ελλάδας και της Ευρωπαϊκής Ένωσης

## Άδειες Χρήσης

- Το παρόν εκπαιδευτικό υλικό υπόκειται σε άδειες χρήσης Creative Commons.
- Για εκπαιδευτικό υλικό, όπως εικόνες, που υπόκειται σε άλλου τύπου άδειας χρήσης, η άδεια χρήσης αναφέρεται ρητώς.



# ΥΠΟΛΟΓΙΣΤΕΣ Ι

## ΣΥΝΑΡΤΗΣΕΙΣ

## Τι είναι μια συνάρτηση;

- Μια ομάδα εντολών, σχεδιασμένη να εκτελεί έναν υπολογισμό και να γυρνάει το αποτέλεσμα
  - Ιδανικές για περιπτώσεις που ο υπολογισμός επαναλαμβάνεται πολλές φορές μέσα στο πρόγραμμα
  - Συντελούν σημαντικά στην καθαρότητα ενός προγράμματος
  - Συντελούν σημαντικά στην μεταβατικότητα ενός προγράμματος

Παράδειγμα: όλες οι «εσωτερικές» συναρτήσεις της Fortran:  
`cos, sin, acos, asin, sqrt, abs, κτλ`

## Παράδειγμα συνάρτησης: υπολογισμός τετραγώνου

- Είναι ήδη γνωστή η εσωτερική συνάρτηση για την τετραγωνική ρίζα `sqrt`.
- Πως θα γράφαμε μια συνάρτηση για το τετράγωνο;
- Επιλέγουμε όνομα που περιγράφει συνοπτικά τι θα κάνει η συνάρτηση: π.χ. `sq`
- Επιλέγουμε τον τύπο των μεταβλητών που δέχεται
  - π.χ. `double precision`
- Επιλέγουμε τον τύπο της μεταβλητής που επιστρέφει
  - εδώ υποχρεωτικά `double precision`

## Δήλωση συνάρτησης `sq`

```
DOUBLE PRECISION FUNCTION SQ ( X )
IMPLICIT NONE
DOUBLE PRECISION X
SQ = X**2
RETURN
END
```

- Γίνεται έξω από το κυρίως πρόγραμμα
  - μετά το `END` του κυρίως προγράμματος
- Χρησιμοποιείται η εντολή `FUNCTION`
  - αριστερά από το `FUNCTION` ο **τύπος επιστροφής**
  - δεξιά από το `FUNCTION` το **όνομα** της συνάρτησης και η **λίστα εισόδου**
- Οι μεταβλητές εισόδου ξαναδηλώνονται
- Το **όνομα της συνάρτησης είναι η μεταβλητή εξόδου**
  - είναι ήδη δηλωμένη: είναι ο τύπος της συνάρτησης
- Η εντολή `RETURN` μας επιστρέφει στο κυρίως πρόγραμμα
  - επιστρέφει την μεταβλητή εξόδου στο κυρίως πρόγραμμα

## Γενική δήλωση συνάρτησης

```

τύπος FUNCTION όνομα (μεταβλητές εισόδου)
IMPLICIT NONE
δηλώσεις μεταβλητών (εισόδου και τοπικών)
εντολές
όνομα = ...
RETURN
END

```

- **τύπος:** π.χ. INTEGER, DOUBLE PRECISION
- **όνομα:** κάτι που περιγράφει τον υπολογισμό που γίνεται
- **μεταβλητές εισόδου:** απο 0 μέχρι και 265 (χωρισμένες με κόμματα) Απλές μεταβλητές ή/και πίνακες. Όποιου τύπου θέλουμε. Αρκεί να έχουν πάρει κάποια τιμή απο το κυρίως πρόγραμμα
- **δηλώσεις μεταβλητών:** οποσδήποτε των μεταβλητών εισόδου, και όποιας τοπικής μεταβλητής δηλώσουμε μέσα στην συνάρτηση
- **εντολές:** ο κώδικας της συνάρτησης
- **όνομα=...:** καθώς το όνομα της συνάρτησης ΕΙΝΑΙ η μεταβλητή εξόδου, πρέπει να της δώσουμε τιμή πριν την έξοδο
- **RETURN:** επιστρέφει την μεταβλητή εξόδου στο κυρίως πρόγραμμα

## Παράδειγμα #1: Συνάρτηση για το μέτρο δισδιάστατου διανύσματος

```

DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
IMPLICIT NONE
DOUBLE PRECISION X, Y

MAGNITUDE = SQRT (X**2 + Y**2)

RETURN
END

```

## Παράδειγμα #2: Συνάρτηση για τον μεγαλύτερο τριών αριθμών

```

DOUBLE PRECISION FUNCTION MAX3 ( X, Y, Z )
IMPLICIT NONE
DOUBLE PRECISION X, Y, Z, M

IF (X .GT. Y) THEN
    M = X
ELSE
    M = Y
END IF

IF (Z .GT. M) THEN
    MAX3 = Z
ELSE
    MAX3 = M
END IF

RETURN
END

```

## Κλήση συνάρτησης

- Καλείται όπως και οι «εσωτερικές» συναρτήσεις
- Μόνη διαφορά: επειδή το όνομα της συνάρτησης είναι και αυτό μεταβλητή, πρέπει να δηλωθεί στο πρόγραμμα που την καλεί

```

PROGRAM TEST_SQ
IMPLICIT NONE
DOUBLE PRECISION X, SQ
WRITE (*,*) 'ΔΩΣΕ ΕΝΑ ΝΟΥΜΕΡΟ'
READ (*,*) X
WRITE (*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', X, 'ΕΙΝΑΙ', SQ(X)
END

DOUBLE PRECISION FUNCTION SQ ( X )
IMPLICIT NONE
DOUBLE PRECISION X
SQ = X**2
RETURN
END

```

## Κλήση συνάρτησης: λεπτομέρεια 1

- Μέσα στην συνάρτηση δεν είμαστε υποχρεωμένοι να διατηρούμε τα ίδια ονόματα για τις μεταβλητές εισόδου, αρκεί οι τύποι να είναι σωστοί, και να δίνονται με την σωστή σειρά

```
PROGRAM TEST_SQ
IMPLICIT NONE
DOUBLE PRECISION X, SQ
WRITE(*,*) 'ΔΩΣΕ ΕΝΑ ΝΟΥΜΕΡΟ'
READ(*,*) X
WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', X, 'ΕΙΝΑΙ', SQ(X)
END

DOUBLE PRECISION FUNCTION SQ ( Y )
IMPLICIT NONE
DOUBLE PRECISION Y
SQ = Y**2
RETURN
END
```

9

## Παράδειγμα #3: Συνάρτηση για το μέτρο δισδιάστατου διανύσματος

```
PROGRAM VECTOR2D
IMPLICIT NONE
DOUBLE PRECISION X1, Y1, MAGNITUDE, M
WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
READ(*,*) X1, Y1

M = MAGNITUDE(X1, Y1)
WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M

END

DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
IMPLICIT NONE
DOUBLE PRECISION X, Y
MAGNITUDE = SQRT (X**2 + Y**2)
RETURN
END
```

10

## Κλήση συνάρτησης: λεπτομέρεια 2

- Εαν η τιμή μιας μεταβλητής εισόδου αλλάξει μέσα στην συνάρτηση, αλλάζει και στο κυρίως πρόγραμμα. **Προσοχή να μην γίνει!!!**

```
PROGRAM VECTOR2D
IMPLICIT NONE
DOUBLE PRECISION X1, Y1, MAGNITUDE, M
WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
READ(*,*) X1, Y1
M = MAGNITUDE(X1, Y1)
WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M
END

DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
IMPLICIT NONE
DOUBLE PRECISION X, Y
MAGNITUDE = SQRT (X**2 + Y**2)
X = 10 * X
RETURN
END
```

11

## Κλήση συνάρτησης: λεπτομέρεια 3

- Μπορούμε να δηλώσουμε νέες τοπικές μεταβλητές στην συνάρτηση. Μετά την επιστροφή χάνονται, δεν επηρεάζουν το κυρίως πρόγραμμα

```
PROGRAM VECTOR2D
IMPLICIT NONE
DOUBLE PRECISION X1, Y1, MAGNITUDE, M
WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
READ(*,*) X1, Y1
M = MAGNITUDE(X1, Y1)
WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M
END

DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
IMPLICIT NONE
DOUBLE PRECISION X, Y, K
K = SQRT (X**2 + Y**2)
MAGNITUDE = K
RETURN
END
```

12

## Παράδειγμα #4: Συνάρτηση για τον υπολογισμό τιμής έκφρασης

Γράψτε συνάρτηση που να γυρνάει την τιμή της έκφρασης:

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

```
DOUBLE PRECISION FUNCTION FUNC1 ( X )
IMPLICIT NONE
DOUBLE PRECISION X, Y

Y = 5*X**2 + 6*X + 7
IF (Y.NE.0) THEN
    FUNC1 = (2*X**2 + 3*X + 4) / Y
ELSE
    FUNC1 = 0
END IF

RETURN
END
```

13

## Παράδειγμα #5: Συνάρτηση για τον υπολογισμό παραγώγου έκφρασης

Γράψτε συνάρτηση που να γυρνάει την τιμή της παραγώγου της παρακάτω συνάρτησης στο  $x=x_0$

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

- Υπάρχουν 2 τρόποι να επιλυθεί
  - Βρίσκουμε αναλυτικά την παράγωγο και παίρνουμε την τιμή της στο  $x=x_0$ .
  - Υπολογίζουμε αριθμητικά την τιμή της παραγώγου στο  $x=x_0$  χρησιμοποιώντας τον ορισμό της παραγώγου:

$$\left. \frac{df(x)}{dx} \right|_{x=x_0} = \lim_{e \rightarrow 0} \frac{f(x+e) - f(x-e)}{2e}$$

Τι να επιλέξουμε για  $e$ ;

14

## Παράδειγμα #5: Συνάρτηση για τον υπολογισμό παραγώγου έκφρασης

- Επιλέγουμε  $e$  πολύ μικρό σχετικά με το  $x_0$ , ας πούμε  $e=x_0/1.000.000$
- Χρησιμοποιούμε την συνάρτηση **FUNC1** που γράψαμε

```
DOUBLE PRECISION FUNCTION DERIV1 ( X )
IMPLICIT NONE
DOUBLE PRECISION X, E, FUNC1

E = X * 1.0D-6
IF (X .EQ. 0) E = 1.0D-6

DERIV = (FUNC1(X+E) - FUNC1(X-E)) / (2*E)


RETURN
END
```

15

## Παράδειγμα #6: Συνάρτηση για τον υπολογισμό ολοκληρώματος έκφρασης

- Το ολοκλήρωμα υπολογίζεται με τον κανόνα του ορθογωνίου

$$\int_{x_1}^{x_2} f(x) dx = \sum_{i=0}^N f(x_i) \Delta x$$



16

## Παράδειγμα #6: Συνάρτηση για τον υπολογισμό ολοκληρώματος έκφρασης

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Γράψτε συνάρτηση που να γυρνάει την τιμή του ολοκληρώματος της παραδίπλα συνάρτησης από  $x_1$  έως  $x_2$  με 10000 διαστήματα

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

```
DOUBLE PRECISION FUNCTION INTEG1 ( X1, X2 )
  IMPLICIT NONE
  INTEGER I, N
  DOUBLE PRECISION X1, X2, X, DX, FUNC1

  N = 10000
  DX = (X2 - X1) / N
  INTEG1 = 0
  DO I = 1, N
    X = (I-1) * DX + DX/2 + X1
    INTEG1 = INTEG1 + FUNC1(X) * DX
  END DO
  RETURN
END
```

7

## Παράδειγμα #7: Συνάρτηση λήψης απόφασης εαν σημείο είναι μέσα ή έξω από κύκλο

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Είσοδος: σημείο, κέντρο κύκλου, ακτίνα κύκλου  
Έξοδος: 1 εαν είναι μέσα, 0 εαν όχι

```
INTEGER FUNCTION INSIDE_CIRCLE (X, Y, X0, Y0, R)
  IMPLICIT NONE
  DOUBLE PRECISION X, Y, X0, Y0, R, D

  D = SQRT( (X-X0)**2 + (Y-Y0)**2 )
  IF (D .LE. R) THEN
    INSIDE_CIRCLE = 1
  ELSE
    INSIDE_CIRCLE = 0
  END IF


  RETURN
END
```

18

## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Έστω ότι δίνονται N σημεία στο επίπεδο. Βρείτε πόσα από αυτά βρίσκονται μέσα σε κάποιον από τους 3 κύκλους, αλλά όχι μέσα σε κοινή τομή τους (δηλαδή πόσα είναι μέσα στην ανοιχτή γαλάζια περιοχή)



Χρησιμοποιήστε την συνάρτηση INSIDE\_CIRCLE

19

## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων (1/2)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

```
PROGRAM POINTS_IN_CIRCLES
  IMPLICIT NONE
  INTEGER NMAX, N, I, INSIDE_CIRCLE, IN(3), SUM
  PARAMETER(NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), X0(3),
  & Y0(3), R(3)

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N
  IF (N .GT. NMAX .OR. N .LE. 0) STOP

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
  READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟΥΣ 3 ΚΥΚΛΟΥΣ'
  READ(*,*) (X0(I), I = 1, 3), (Y0(I), I = 1, 3),
  & (R(I), I = 1, 3)
```

ΣΥΝΕΧΙΖΕΤΑΙ...

20

## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων (2/2)

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ

```
SUM = 0

DO I = 1, N
  IN(1)=INSIDE_CIRCLE(X(I),Y(I),X0(1),Y0(1),R(1))
  IN(2)=INSIDE_CIRCLE(X(I),Y(I),X0(2),Y0(2),R(2))
  IN(3)=INSIDE_CIRCLE(X(I),Y(I),X0(3),Y0(3),R(3))
  IF (IN(1)+IN(2)+IN(3) .EQ. 1) SUM = SUM + 1
END DO

WRITE(*,*) 'ΜΕΣΑ ΣΤΟΥΣ ΚΥΚΛΟΥΣ ΕΙΝΑΙ', SUM, 'ΣΗΜΕΙΑ'


END
```

21

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων

Έστω ότι δίνονται N σημεία στο επίπεδο. Βρείτε πόσα από αυτά βρίσκονται μέσα σε οποιαδήποτε από τις τομές των M κύκλων (δηλαδή πόσα είναι μέσα στις σκούρες γαλάζιες περιοχές)

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ



22

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων (1/2)

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ

```
PROGRAM POINTS_IN_CIRCLES
IMPLICIT NONE
INTEGER NMAX, N, M, I, J, INSIDE_CIRCLE, IN, SUM
PARAMETER(NMAX = 1000)
DOUBLE PRECISION X(NMAX), Y(NMAX), X0(NMAX),
&                Y0(NMAX), R(NMAX)

WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ; ΚΥΚΛΟΥΣ;'
READ(*,*) N, M
IF (N .GT. NMAX .OR. M .GT. NMAX) STOP

WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟΥΣ 3 ΚΥΚΛΟΥΣ'
READ(*,*) (X0(I), I = 1, M), (Y0(I), I = 1, M),
&         (Y(I), I = 1, M)

ΣΥΝΕΧΙΖΕΤΑΙ...
```

23

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων (2/2)

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ

```
SUM = 0

DO I = 1, N
  IN = 0
  DO J = 1, M
    IN = IN + INSIDE_CIRCLE(X(I),Y(I),
&                          X0(J),Y0(J),R(J))
  END DO

  IF (IN .GE. 2) SUM = SUM + 1
END DO

WRITE(*,*) 'ΜΕΣΑ ΣΤΟΥΣ ΚΥΚΛΟΥΣ ΕΙΝΑΙ', SUM, 'ΣΗΜΕΙΑ'

END
```

24

## Εισαγωγή πινάκων σε συναρτήσεις: π.χ. μέτρο δισδιάστατου διανύσματος

```
PROGRAM VECTOR2D
IMPLICIT NONE
DOUBLE PRECISION P(2), MAGNITUDE, M
WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
READ(*,*) P(1), P(2)

M = MAGNITUDE(P)
WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', P(1), P(2), 'ΕΙΝΑΙ', M

END

DOUBLE PRECISION FUNCTION MAGNITUDE ( P )
IMPLICIT NONE
DOUBLE PRECISION P(2)
MAGNITUDE = SQRT ( P(1)**2 + P(2)**2 )
RETURN
END
```

25

## Εισαγωγή πινάκων σε συναρτήσεις

- Στην λίστα μεταβλητών εισόδου μπαίνει ΜΟΝΟ το όνομα του πίνακα
- Στην δήλωση μεταβλητών εισόδου δηλώνουμε το μέγεθος του πίνακα:
- Το μέγεθος του πίνακα μπορεί να περάσει και αυτό σαν είσοδος

26

## Παράδειγμα #10: Μέτρο διανύσματος N διαστάσεων

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το μέτρο του

```
DOUBLE PRECISION FUNCTION MAGNITUDE_N ( X, N )
IMPLICIT NONE
INTEGER N, I
DOUBLE PRECISION X(N), S

S = 0
DO I = 1, N
    S = S + X(I)**2
END DO

MAGNITUDE_N = SQRT ( S )

RETURN
END
```

27

## Γιατί εισάγουμε μόνο το πλήθος των ενεργών στοιχείων και όχι την μέγιστη διάσταση του πίνακα;

- Η σωστή ερώτηση είναι: όταν εισάγουμε έναν πίνακα σε μια συνάρτηση, εισάγουμε όλο τον πίνακα (δηλαδή όλα τα στοιχεία του);
- Η απάντηση είναι ΟΧΙ: κατά την είσοδο μέσα στην συνάρτηση μεταφέρεται μόνο η διεύθυνση μνήμης του πρώτου στοιχείου του πίνακα, τίποτα άλλο (*κλήση κατά αναφορά*). Κατόπιν η προσπάθεια γίνεται σειριακά.
- Άρα η δήλωση της μέγιστης διάστασης δεν είναι υποχρεωτική όταν έχουμε μονοδιάστατο πίνακα
- Είναι υποχρεωτική όταν έχουμε πολυδιάστατο πίνακα

28



## Παράδειγμα #11: Ελάχιστο πίνακα

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το ελάχιστο στοιχείο

```
DOUBLE PRECISION FUNCTION XMIN (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N)

  XMIN = X(1)
  DO I = 2, N
    IF (X(I) .LT. XMIN) XMIN = X(I)
  END DO

  RETURN
END
```

29

## Παράδειγμα #12: Μέσος όρος πίνακα

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει τον μέσο όρο

```
DOUBLE PRECISION FUNCTION AVERAGE (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N)

  AVERAGE = 0
  DO I = 1, N
    AVERAGE = AVERAGE + X(I)
  END DO
  AVERAGE = AVERAGE / N
  RETURN
END
```

30

## Παράδειγμα #13: Σφάλμα μέσου όρου

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το σφάλμα του μέσου όρου

- Σε ένα σύνολο μετρήσεων  $x_i$ ,  $i=1, N$ , υπολογίζουμε

μέσος όρος

$$\bar{X} = \frac{\sum_{i=1}^N X_i}{N}$$

Σφάλμα μέσου όρου

$$\sigma_x = \left[ \frac{\sum_{i=1}^N (x_i - \bar{x})^2}{N(N-1)} \right]^{1/2}$$

31

## Παράδειγμα #13: Σφάλμα μέσου όρου

$$\sigma_{\bar{x}} = \left[ \frac{\sum_{i=1}^N (x_i - \bar{x})^2}{N(N-1)} \right]^{1/2}$$

```
DOUBLE PRECISION FUNCTION STD (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N), XMEAN, AVERAGE

  XMEAN = AVERAGE(X, N)

  STD = 0
  DO I = 1, N
    STD = STD + (X(I) - XMEAN)**2
  END DO

  STD = SQRT( STD / (N*(N-1) ))

  RETURN
END
```

32

## Παράδειγμα #14 Στατιστική ανάλυση

Κατασκευάστε πρόγραμμα που δέχεται N αριθμούς και θα κάνει στατιστική ανάλυση

```
PROGRAM STATISTICS
IMPLICIT NONE
INTEGER NMAX, N, I
PARAMETER(NMAX = 1000)
DOUBLE PRECISION X(NMAX), AVERAGE, STD

WRITE(*,*) 'ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ'
READ(*,*) N
IF (N .GT. NMAX .OR. N .LE. 0) STOP

WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ'
READ(*,*) (X(I), I = 1, N)

WRITE(*,*) AVERAGE(X, N), '+-', STD(X, N)
END
```

3

## Παράδειγμα #15: Ελάχιστο πίνακα μεταξύ N1 και N2

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και δύο ακέραιους N1 και N2, και να επιστρέφει το ελάχιστο στοιχείο μεταξύ N1 και N2

```
DOUBLE PRECISION FUNCTION XMIN (X, N1, N2)
IMPLICIT NONE
INTEGER N1, N2, I
DOUBLE PRECISION X(N2)

XMIN = X(N1)
DO I = N1+1, N2
    IF(X(I).LT.XMIN) XMIN = X(I)
END DO

RETURN
END
```

34

## Παράδειγμα #16: Θέση ελαχίστου πίνακα

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και να επιστρέφει την θέση στην οποία βρίσκεται το ελάχιστο στοιχείο του

Λύση:

- Η συνάρτηση θα δέχεται έναν μονοδιάστατο πίνακα καθώς και τον αριθμό των ενεργών στοιχείων του
- Μέσα στην συνάρτηση βρίσκουμε και το ελάχιστο στοιχείο αλλά και την θέση στην οποία βρίσκεται αυτό, και γυρνάμε πίσω την θέση
- **Συνάρτηση τύπου INTEGER**

35

## Παράδειγμα #16: Θέση ελαχίστου πίνακα

```
INTEGER FUNCTION IMIN (X, N)
IMPLICIT NONE
INTEGER N, I
DOUBLE PRECISION X(N), XMIN

XMIN = X(1)
IMIN = 1
DO I = 2, N
    IF(X(I).LT.XMIN) THEN
        XMIN = X(I)
        IMIN = I
    END IF
END DO

RETURN
END
```

36

## Εισαγωγή δισδιάστατου πίνακα

- Εδώ πρέπει να εισάγουμε τουλάχιστον το μέγιστο πλήθος γραμμών του πίνακα
- ΔΕΝ μπορούμε να εισάγουμε μόνο το πλήθος των ενεργών στοιχείων όπως κάνουμε με τους μονοδιάστατους πίνακες
- Στην γενική περίπτωση με πίνακες  $M$  διαστάσεων, δηλώνουμε επακριβώς τις πρώτες  $M-1$  διαστάσεις

37

## Παράδειγμα #17: ίχνος δισδιάστατου πίνακα

Γράψτε συνάρτηση που να δέχεται έναν δισδιάστατο τετραγωνικό πίνακα και να επιστρέφει το ίχνος του

```
DOUBLE PRECISION FUNCTION TRACE (X, N)
IMPLICIT NONE
INTEGER N, I
DOUBLE PRECISION X(N, N)

TRACE = 0
DO I = 1, N
    TRACE = TRACE + X(I, I)
END DO

RETURN
END
```

Εδώ το  $N$  πρέπει αναγκαστικά να είναι η μέγιστη διάσταση

38

## Παράδειγμα #18: άθροισμα των $N$ πρώτων στοιχείων της διαγωνίου

Γράψτε συνάρτηση που να δέχεται έναν δισδιάστατο τετραγωνικό πίνακα και να επιστρέφει το άθροισμα των  $N$  πρώτων στοιχείων της διαγωνίου του

```
DOUBLE PRECISION FUNCTION TRACE (X, NMAX, N)
IMPLICIT NONE
INTEGER NMAX, I
DOUBLE PRECISION X(NMAX, N)

TRACE = 0
DO I = 1, N
    TRACE = TRACE + X(I, I)
END DO

RETURN
END
```

Εδώ πρέπει αναγκαστικά να εισάγουμε και τη μέγιστη διάσταση

## Γιατί αυτή η διαφορά μεταξύ μονοδιάστατων και πολυδιάστατων;

έστω ένας πίνακας  $X(9)$ , και θέλουμε να επεξεργαστούμε τα 4 πρώτα στοιχεία του

ο πίνακας  $X(9)$ , όπως είναι στο κυρίως πρόγραμμα

(1  
2  
3  
4  
5  
6  
7  
8  
9)

εάν τον περάσουμε σε συνάρτηση και καλέσουμε τα 4 πρώτα στοιχεία του, θα έχουμε

(1  
2  
3  
4)

## Γιατί αυτή η διαφορά μεταξύ μονοδιάστατων και πολυδιάστατων;

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ

έστω τώρα ένας πίνακας  $X(3,3)$ , και θέλουμε να επεξεργαστούμε το πρώτο  $(2,2)$  τεταρτημόριό του

ο πίνακας  $X(3,3)$ , όπως είναι στο κυρίως πρόγραμμα

$$\begin{pmatrix} 11 & 12 & 13 \\ 21 & 22 & 23 \\ 31 & 32 & 33 \end{pmatrix}$$

ο πίνακας  $X$ , όπως είναι στη μνήμη

$$\begin{pmatrix} 11 \\ 21 \\ 31 \\ 12 \\ 22 \\ 32 \\ 13 \\ 23 \\ 33 \end{pmatrix}$$

στην συνάρτηση, εαν τον δηλώσουμε ως  $2 \times 2$  πίνακα, να τι θα πάρουμε

$$\begin{pmatrix} 11 & 31 \\ 21 & 12 \end{pmatrix} \text{ λάθος}$$

εαν όμως τον δηλώσουμε ως  $3 \times 2$ , θα πάρουμε

$$\begin{pmatrix} 11 & 12 \\ 21 & 22 \\ 31 & 32 \end{pmatrix} \text{ σωστό!!}$$

## Τέλος Ενότητας



## Χρηματοδότηση

- Το παρόν εκπαιδευτικό υλικό έχει αναπτυχθεί στα πλαίσια του εκπαιδευτικού έργου του διδάσκοντα.
- Το έργο «**Ανοικτά Ακαδημαϊκά Μαθήματα στο Πανεπιστήμιο Ιωαννίνων**» έχει χρηματοδοτήσει μόνο τη αναδιαμόρφωση του εκπαιδευτικού υλικού.
- Το έργο υλοποιείται στο πλαίσιο του Επιχειρησιακού Προγράμματος «**Εκπαίδευση και Δια Βίου Μάθηση**» και συγχρηματοδοτείται από την Ευρωπαϊκή Ένωση (Ευρωπαϊκό Κοινωνικό Ταμείο) και από εθνικούς πόρους.



Ευρωπαϊκή Ένωση  
Ευρωπαϊκό Κοινωνικό Ταμείο



ΕΠΙΧΕΙΡΗΣΙΑΚΟ ΠΡΟΓΡΑΜΜΑ  
ΕΚΠΑΙΔΕΥΣΗ ΚΑΙ ΔΙΑ ΒΙΟΥ ΜΑΘΗΣΗ  
Περίοδος προγραμματισμού 2007-2013  
ΥΠΟΥΡΓΕΙΟ ΠΑΙΔΕΙΑΣ & ΘΡΗΣΚΕΥΜΑΤΩΝ, ΠΟΛΙΤΙΣΜΟΥ & ΑΘΛΗΤΙΣΜΟΥ  
ΕΙΔΙΚΗ ΥΠΗΡΕΣΙΑ ΔΙΑΧΕΙΡΙΣΗΣ  
Με τη συγχρηματοδότηση της Ελλάδας και της Ευρωπαϊκής Ένωσης



ΕΣΠΑ  
2007-2013  
Μίσθια για το μέλλον  
ΕΥΡΩΠΑΪΚΟ ΚΟΙΝΩΝΙΚΟ ΤΑΜΕΙΟ

## Σημειώματα

### Σημείωμα Ιστορικού Εκδόσεων Έργου

Το παρόν έργο αποτελεί την έκδοση 1.0.

Έχουν προηγηθεί οι κάτωθι εκδόσεις:

- Έκδοση 1.0 διαθέσιμη εδώ.

<http://ecourse.uoi.gr/course/view.php?id=1154>.

## Σημείωμα Αναφοράς

Copyright Πανεπιστήμιο Ιωαννίνων, Διδάσκοντες: Αν.  
Καθ. Δ. Παπαγεωργίου, Αν. Καθ. Ε. Λοιδωρίκης.  
«Υπολογιστές Ι. Συναρτήσεις». Έκδοση: 1.0. Ιωάννινα  
2014. Διαθέσιμο από τη δικτυακή διεύθυνση:  
<http://ecourse.uoi.gr/course/view.php?id=1154>.

## Σημείωμα Αδειοδότησης

- Το παρόν υλικό διατίθεται με τους όρους της άδειας χρήσης Creative Commons Αναφορά Δημιουργού - Παρόμοια Διανομή, Διεθνής Έκδοση 4.0 [1] ή μεταγενέστερη.



- [1] <https://creativecommons.org/licenses/by-sa/4.0/>.